G08AEF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

G08AEF performs the Friedman two-way analysis of variance by ranks on k related samples of size n.

2 Specification

```
SUBROUTINE G08AEF(X, IX, K, N, W1, W2, FR, P, IFAIL)
INTEGER           IX, K, N, IFAIL
real              X(IX,N), W1(K), W2(K), FR, P
```

3 Description

The Friedman test investigates the score differences between k matched samples of size n, the scores in the ith sample being denoted by:

$$x_{i1}, x_{i2}, \ldots, x_{in}.$$

(Thus the sample scores may be regarded as a two-way table with k rows and n columns.) The hypothesis under test, H_0, often called the null hypothesis, is that the samples come from the same population, and this is to be tested against the alternative hypothesis H_1 that they come from different populations.

The test is based on the observed distribution of score rankings between the matched observations in different samples.

The test proceeds as follows:

(a) The scores in each column are ranked, r_{ij} denoting the rank within column j of the observation in row i. Average ranks are assigned to tied scores.

(b) The ranks are summed over each row to give rank sums $t_i = \sum_{j=1}^{n} r_{ij}$, for $i = 1, 2, \ldots, k$.

(c) The Friedman test statistic FR is computed, where

$$FR = \frac{12}{nk(k+1)} \sum_{i=1}^{k} \{t_i - \frac{1}{2}n(k+1)\}^2.$$

G08AEF returns the value of FR, and also an approximation, p, to the significance of this value. (FR approximately follows a χ^2_{k-1} distribution, so large values of FR imply rejection of H_0). H_0 is rejected by a test of chosen size α if $p < \alpha$. The approximation p is acceptable unless $k = 4$ and $n < 5$, or $k = 3$ and $n < 10$, or $k = 2$ and $n < 20$; for $k = 3$ or 4, tables should be consulted (e.g., N of Seigel [1]); for $k = 2$ the Sign test (see G08AAF) or Wilcoxon test (see G08ABF) is in any case more appropriate.

4 References

[1] Siegel S (1956) *Non-parametric Statistics for the Behavioral Sciences* McGraw–Hill

5 Parameters

1: X(IX,N) — ***real*** array *Input*

On entry: X(i, j) must be set to the value, x_{ij}, of observation j in sample i, for $i = 1, 2, \ldots, k$; $j = 1, 2, \ldots, n$.

2: IX — INTEGER *Input*

On entry: the first dimension of the array X as declared in the (sub)program from which G08AEF is called.

Constraint: IX $\geq$ K.

3: K — INTEGER *Input*

On entry: the number of samples, k.

Constraint: K > 1.

4: N — INTEGER *Input*

On entry: the size, n, of each sample.

Constraint: N ≥ 1.

5: W1(K) — ***real*** array *Workspace*

6: W2(K) — ***real*** array *Workspace*

7: FR — ***real*** *Output*

On exit: the value of the Friedman test statistic, FR.

8: P — ***real*** *Output*

On exit: the approximate significance, p, of the Friedman test statistic.

9: IFAIL — INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. For users not familiar with this parameter (described in Chapter P01) the recommended value is 0.

On exit: IFAIL $= 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors detected by the routine:

IFAIL $= 1$

On entry, N < 1.

IFAIL $= 2$

On entry, IX $<$ K.

IFAIL $= 3$

On entry, K ≤ 1.

7 Accuracy

For estimates of the accuracy of the significance p, see G01ECF. The χ^2 approximation is acceptable unless $k = 4$ and $n < 5$, or $k = 3$ and $n < 10$, or $k = 2$ and $n < 20$.

8 Further Comments

The time taken by the routine is approximately proportional to the product nk.

If $k = 2$, the Sign test (see G08AAF) or Wilcoxon test (see G08AGF) is more appropriate.

9 Example

This example is taken from page 169 of Siegel [1]. The data relate to training scores of three matched samples of 18 rats, trained under three different patterns of reinforcement.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     G08AEF Example Program Text
*     Mark 14 Revised.  NAG Copyright 1989.
*     .. Parameters ..
      INTEGER          K, N, IX
      PARAMETER        (K=3,N=18,IX=K)
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
*     .. Local Scalars ..
      real             FR, SIG
      INTEGER          I, IFAIL, J
*     .. Local Arrays ..
      real             W1(K), W2(K), X(IX,N)
*     .. External Subroutines ..
      EXTERNAL         G08AEF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'G08AEF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) ((X(I,J),J=1,18),I=1,3)
      WRITE (NOUT,*)
      WRITE (NOUT,*) 'Friedman test'
      WRITE (NOUT,*)
      WRITE (NOUT,*) 'Data values'
      WRITE (NOUT,*)
      WRITE (NOUT,*) '   Group Group Group'
      WRITE (NOUT,*) '     1     2     3'
      WRITE (NOUT,99997) ((X(I,J),I=1,3),J=1,18)
      IFAIL = 0
*
      CALL G08AEF(X,IX,K,N,W1,W2,FR,SIG,IFAIL)
*
      WRITE (NOUT,*)
      WRITE (NOUT,99999) 'Test statistic       ', FR
      WRITE (NOUT,99998) 'Degrees of freedom   ', K - 1
      WRITE (NOUT,99999) 'Significance         ', SIG
      STOP
*
99999 FORMAT (1X,A,F6.3)
99998 FORMAT (1X,A,I6)
99997 FORMAT (1X,F7.1,2F6.1)
      END
```

9.2 Program Data

```
G08AEF Example Program Data
  1  2  1  1  3  2  3  1  3  3  2  2  3  2  2.5  3  3  2
  3  3  3  2  1  3  2  3  1  1  3  3  2  3  2.5  2  2  3
  2  1  2  3  2  1  1  2  2  2  1  1  1  1  1  1  1  1
```

9.3 Program Results

```
G08AEF Example Program Results

Friedman test

Data values

   Group Group Group
     1     2     3
    1.0   3.0   2.0
    2.0   3.0   1.0
    1.0   3.0   2.0
    1.0   2.0   3.0
    3.0   1.0   2.0
    2.0   3.0   1.0
    3.0   2.0   1.0
    1.0   3.0   2.0
    3.0   1.0   2.0
    3.0   1.0   2.0
    2.0   3.0   1.0
    2.0   3.0   1.0
    3.0   2.0   1.0
    2.0   3.0   1.0
    2.5   2.5   1.0
    3.0   2.0   1.0
    3.0   2.0   1.0
    2.0   3.0   1.0

Test statistic        8.583
Degrees of freedom        2
Significance          0.014
```